

# **A Tutorial on Using EXCEL and EXCEL Add-ins to Value Real Options**

**By Wayne L. Winston**

**Professor of Decision Sciences**

**Kelley School of Business**

**Bloomington IN, 47405**

**April 15, 1999**

## Section 1-Introduction

EXCEL and EXCEL add-ins can greatly simplify the solution of many real option problems. We will give several examples of how EXCEL and the add-ins @RISK (a simulation add-in) and RISKOPTIMIZER (a simulation and optimization add-in) can be used to quickly and easily solve many real options problems. We then conclude with a new, easily implementable approach to option pricing that **does not require the underlying asset to be a Lognormal random variable**. In Section 2 we begin with a brief discussion of the risk neutral valuation approach. In Section 3 we introduce the Lognormal random variable and show how Lognormal parameters can easily be estimated from historical data or implied volatility. We also show how to use @RISK to simulate a Lognormal random variable. Section 4 illustrates the use of simulation to price a European put (the equivalent of an abandonment option). Section 5 shows how the simulation approach simplifies the valuation of many European real options such as expansion, contraction and abandonment that are discussed in Trigeorgis (1996). Section 6 shows how to price American options with Excel. Optimal exercise points are highlighted using EXCEL's Conditional Formatting Feature. Section 6 also shows how simulation can be used to obtain the distribution of cash flows for an American option. Sections 7 and 8 show how to use the EXCEL-add in RISKOPTIMIZER to price more complex American options, including an option to start up and close a gold mine. In Section 9 we show how RISKOPTIMIZER can be used to model the decision to enter a new market. Basically, this example shows how a decision tree involving continuous random variables such as market share and market growth may be modeled via simulation. The material in Sections 2-9 has appeared in Winston (1998) and Winston (1999). In Section 10 we show the ideas of Luenberger (1997) can be extended to easily price financial and real options for which the underlying asset follows **any distribution**. In particular, the distribution of the underlying asset may be obtained from historical data or from simulation. In Section 11 we show how to value an option to cancel an order for airplanes.

## Section 2-The Risk Neutral Approach to Option Pricing

A **European option** on a stock gives the owner of the option the right to buy (if the option is a **call option**) or sell (if the option is a **put option**) one share of stock for a particular price (**the exercise price**) on a particular date (**the exercise date**). An **American option** allows you buy or sell the stock at any date between the present and the exercise date.

Options are usually priced by arbitrage arguments. For example, suppose 3 months from now a stock will sell for either \$18 (bad state) or \$22 (good state). The stock currently sells for \$20 and we own a 3-month European call option with an exercise price of \$21. The risk free rate is 12% per year. Three months from now the option is worth \$1 (in good state) or \$0 (in bad state). If we create a portfolio that is long .25

shares of stock and short 1 call we will show that in both Good and Bad states this portfolio yields \$4.50.

State	Portfolio Value
Good	$.25(\$22) - (1)(\$1) = \$4.50.$
Bad	$.25(\$18) - (1)(\$0) = \$4.50.$

Note that in the Good state the option is worth \$1 because we can exercise the option and buy the stock for \$20 and immediately sell the stock for \$21. In the Bad state the option is worthless because there is no benefit to be gained by buying the stock for \$20 when the current price is \$18!

Since this portfolio yields \$4.50 for certain three months from now its value today must equal the NPV (discounted at the risk-free rate) of \$4.50 received three months from now. This is just

$$4.5e^{-.25(.12)} = \$4.37. \text{ This implies that}$$

$$.25(\text{today's stock price}) - (\text{today's option price}) = \$4.37 \text{ or}$$

$$5 - (\text{today's option price}) = \$4.37 \text{ or}$$

$$(\text{today's option price}) = \$0.63.$$

This pricing approach is called **arbitrage pricing**. The argument works independent of a person's risk preferences. Therefore we may use the following approach to price derivatives;

1. In a world where everyone is risk neutral arbitrage pricing is valid.
2. In a risk neutral world all assets must grow on average at risk free rate.
3. In a risk neutral world any asset (including an option) is worth the expected value of its discounted cash flows.
4. Set up a risk neutral world in which all stocks grow at risk free rate and use @ RISK (or binomial tree; see Chapters 6-58) to determine expected discounted cash flows from option.
5. **Since the arbitrage pricing method gives the correct price in all worlds it yields the correct price in a risk neutral world. Therefore if we use the above method to find a derivative's price in a risk neutral world we have found the right price for our complicated non risk-neutral world!**

It is important to note that actual growth rate of a stock is irrelevant to pricing a derivative. Information about a stock's future growth rate is imbedded in today's stock price.

### **Example of Risk Neutral Approach**

Let's find the call option value of \$0.63 using the risk neutral approach. Let  $p$  be probability (in the risk neutral world) that in 3 months stock price is \$22. Then  $1 - p$  is probability that stock price 3 months from now is \$18. If stock grows on average at risk free rate we must have

$$p(\$22) + (1-p)(\$18) = 20 * e^{(.25)(.12)} = \$20.61 \text{ or}$$

$$4p = 2.61$$

or

$$p = .65.$$

Now value option as expected discounted value of its cash flows:

$$e^{-.25(.12)}(.65(\$1) + .35(\$0)) = \$0.63.$$

For further discussion of the risk neutral approach we refer the reader to Hull (1997).

### **Section 3-The Lognormal Model of Stock Prices**

The Lognormal model for asset value (or stock price) assumes that in a small time  $\Delta t$  the stock price changes by an amount that is normally distributed with

$$\text{Standard Deviation} = \sigma S \sqrt{\Delta t}$$

$$\text{Mean} = \mu S \Delta t$$

Here  $S$  = current stock price.

$\mu$  may be thought of as the instantaneous rate of return on the stock. By the way, this model leads to really "jumpy" changes in stock prices (like real life). This is because during a small period of time the standard deviation of the stock's movement will greatly exceed the mean of a stock's movement. This follows because for small  $\Delta t$ ,  $\sqrt{\Delta t}$  will be much larger than  $\Delta t$ .

In a small time  $\Delta t$  the natural logarithm ( $\ln(S)$ ) of the current stock price will change (by Ito's Lemma, see Hull (1997)) by an amount that is normally distributed with

$$\text{Mean} = (\mu - .5\sigma^2)\Delta t$$

$$\text{Standard Deviation} = \sigma \sqrt{\Delta t}$$

Let  $S_t$  = stock price at time  $t$ . In Chapter 11 of Hull (1997) it is shown that at time  $t$   $\ln S_t$  is normally distributed with

$$\text{Mean} = \ln S_0 + (\mu - .5\sigma^2)t \text{ and}$$

$$\text{Standard Deviation} = \sigma\sqrt{t}$$

We refer to  $(\mu - .5\sigma^2)$  as the **continuously compounded rate of return** on the stock. Note the continuously compounded rate of return on  $S$  is less than instantaneous return. Since  $\ln S_t$  follows a normal random variable we say that  $S_t$  is a **Lognormal random variable**.

To simulate  $S_t$  we get  $\ln(S_t)$  by entering in @RISK the formula

$$= \text{LN}(S_0) + (\mu - .5\sigma^2)t + \sigma\sqrt{t}\text{RISKNORMAL}(0,1)$$

Therefore to get  $S_t$  we must take the antilog of this equation and get

$$S_t = S_0 e^{(\mu - .5\sigma^2)t + \sigma\sqrt{t}\text{RISKNORMAL}(0,1)} \quad (2.1).$$

### **Risk Neutral Valuation**

To apply the risk neutral valuation approach of Section 2 we assume the asset grows at instantaneous rate  $r$ . **Then the value of a derivative is simply the expected discounted (at risk free rate) value of cash flows.** We will often apply this approach. We can estimate volatility by implied volatility (see Section 3) or historical volatility (see below)

### **Historical Estimation of Mean and Volatility of Stock Return**

If we average values of

$$\ln \frac{S_t}{S_{t-1}}$$

we obtain an estimate of  $(\mu - .5\sigma^2)$ .

If we take the standard deviation of

$$\ln \frac{S_t}{S_{t-1}}$$

we obtain an estimate of  $\sigma$ .

Using monthly returns of Dell Computer for 1988-1996 we may estimate  $\mu$  and  $\sigma$ . See the file Dell.xls.

Figure 3.1

	A	B	C	D	E	F
1		estimate of	monthly			
2		$\mu-.5*\sigma^2$	0.034572			
3		sigma	0.160929			
4	<b>DATE</b>	<b>Dell</b>	<b>Ln(1+Dell)</b>			
5	6/30/88				monthly estimate	
6	7/29/88	0.106667	0.101353		sigma	0.160929
7	8/31/88	-0.228916	-0.25996		mu	0.047521
8	9/30/88	0.28125	0.247836		annual estimate	
9	10/31/88	0.158537	0.147158		sigma	0.557473
10	11/30/88	-0.0842105	-0.08797		mu	0.570254
11	12/30/88	-0.0804598	-0.08388			
12	1/31/89	-0.05	-0.05129			
13	2/28/89	-0.171053	-0.1876			
14	3/31/89	-0.0952381	-0.10008			
15	4/28/89	0.105263	0.100083			
16	5/31/89	0.0793651	0.076373			
17	6/30/89	-0.0735294	-0.07637			
18	7/31/89	-0.142857	-0.15415			
19	8/31/89	0.037037	0.036368			
20	9/29/89	0.0178571	0.0177			
21	10/31/89	-0.157895	-0.17185			
22	11/30/89	-0.0416667	-0.04256			
23	12/29/89	-0.0434783	-0.04445			
24	1/31/90	-0.159091	-0.17327			
25	2/28/90	0.351351	0.301105			
26	3/30/90	0.22	0.198851			
27	4/30/90	0.114754	0.108634			
99	4/30/96	0.369403	0.314375			
100	5/31/96	0.207084	0.188208			
101	6/28/96	-0.0812641	-0.08476			
102	7/31/96	0.0909091	0.087011			
103	8/30/96	0.209459	0.190173			
104	9/30/96	0.158287	0.146942			
105	10/31/96	0.0466238	0.04557			
106	11/29/96	0.248848	0.222222			
107	12/31/96	0.0455105	0.044505			

We begin by estimating  $\sigma$  and  $\mu$  for a monthly Lognormal process. Then our estimate of  $\sigma$  for an annual Lognormal process is just  $\sqrt{12}$  (monthly estimate of  $\sigma$ ) and our estimate of  $\mu$  for an annual Lognormal process is just  $12*(\text{monthly estimate of } \mu)$ .

**Step 1:** Note that for any month  $\frac{S_{t+1}}{S_t} = (1 + \text{Montht Return})$ . Therefore in C6:C107 we

compute for each month  $\text{Ln} \frac{S_{t+1}}{S_t}$  by copying from C6 to C7:C107 the formula

$$=\text{LN}(1+\text{B6}).$$

**Step 2:** In C2 we estimate  $\mu - .5\sigma^2$  with the formula

$$=\text{AVERAGE}(C6:C107).$$

**Step 3:** In C3 we estimate  $\sigma$  with the formula

$$=\text{STDEV}(C6:C107).$$

**Step 4:** In cell F7 we find our estimate of  $\mu$  for a monthly lognormal process with the formula

$$=C2+0.5*\text{F6}^2.$$

Thus for the monthly Lognormal we estimate for DELL Computer that  $\mu = .0475$  and  $\sigma = .161$ .

**Step 5:** In cells F9 and F10 we find our annualized estimates of  $\mu$  and  $\sigma$  with the formulas

$$=12*\text{F7} \text{ (for } \mu \text{)}$$

$$=\text{SQRT}(12)*\text{F6}. \text{ (for } \sigma \text{)}$$

Our annualized estimates are  $\mu = 57.0\%$  and  $\sigma = 5.7\%$ .

### ***Finding Mean and Variance of a Lognormal Random Variable***

It is important to point out that  $\mu$  is not actually the mean of a Lognormal random variable and  $\sigma$  is not really the standard deviation. Assume a stock follows a Lognormal random variable with parameters  $\mu$  and  $\sigma$ . Let  $S$  = current price of stock (which is known) and  $S_t$  = Price of stock at time  $t$  (unknown). Then (see page 310 of Luenberger (1997)) the mean and variance of  $S_t$  are as follows:

$$\text{Mean of } S_t = Se^{\mu t}$$



$$\text{Variance of } S_t = S^2 e^{2\mu t} (e^{\sigma^2 t} - 1).$$

The file Lognormal.xls contains a template to determine the mean and variance of a stock price at any future time. See Figure 3.2

**Figure 3.2**

	A	B	C	D
2				
3	S=current price	20		
4	t=time	1		
5	mu	0.2		
6	sigma	0.4		
7	alpha	0.95		
8				
9	Mean for ln S(T)	3.115732	Mean S(T)	24.42806
10	Sigma for ln S(T)	0.4	var S(T)	103.5391
11			sigma S(T)	10.17542
12	<b>CI</b>			
13	Lower(for ln S(T))	2.331748		
14	Upper (for ln S(T))	3.899717		
15	Lower for S(T)	10.29592		
16	Upper for S(T)	49.38846		

For example, consider a Stock currently selling for \$20 following a Lognormal with  $\mu = .20$  and  $\sigma = .40$ . The mean stock price one-year from now is \$24.43 with a standard deviation of \$10.18.

### **Confidence Intervals for a Lognormal Random Variable**

The file Lognormal.xls computes a confidence interval for a future stock price. If you want a 95% Confidence interval enter .95 for alpha, etc. From Figure 41.2 we find that for a stock currently selling for \$20 with  $\mu = .20$  and  $\sigma = .40$ , we are 95% sure that the price of the stock one year from now will be between \$10.30 and \$49.39.

## **Section 4-Pricing European Puts and Calls by Simulation**

Even though European puts and calls can easily be priced by the Black-Scholes (BS) formula, it is instructive to use Monte Carlo simulation to price European options. The file option.xls contains a template that gives the BS price for a European put or call. Of course, the key input into pricing an option is the stock's **volatility**. This is just the value of  $\sigma$  in the Lognormal representation of the stock's price. In Section 3 we showed how historical data could be used to estimate  $\sigma$ . **Implied volatility** is a more commonly used method for estimating a stock's volatility. Given the price of an option, the stock's implied volatility is the value of  $\sigma$  which makes the BS price for the option match the actual price. In a sense the volatility estimate is "implied" by the actual option price. We now show how to use GOAL SEEK to find an implied volatility. We note that the risk

free rate input to the BS formula should be the continuously compounded rate, or  $\ln(1 + \text{current 90 day T-Bill Rate})$ .

### Example 4.1

On June 30, 1998 DELL Computer sold for \$94. A European put with an exercise price of \$80 expiring on November 22, 1998 was selling for \$5.25. The current 90 day T-Bill rate is 5.5%. What is the implied volatility of DELL computer?

### Solution


Our work is in the file option.xls. See Figure 4.1

Figure 4.1

	A	B	C	D	E	F
1	<b>Black-Schole's Option Pricing Problem</b>					
2	<b>Using the Option Price to Find the Implied Volatility</b>					
3						
4	<b>Input data</b>					
5	Stock price	\$94			today	6/30/98
6	Exercise price	\$80			expire	11/22/98
7	Duration	0.39726				
8	Interest rate	5.35%				
9						
10	Implied volatility (stdev)	53.27%				
11						
12	Call price	Actual		Predicted		
13			=	\$21.06		
14	Put price	\$5.25		\$5.25		
15						
16						
17	Other quantities for option price					
18	d1	0.715534		N(d1)	0.76286	
19	d2	0.37981		N(d2)	0.647957	

**Step 1:** Enter the duration of the option (145/365) years in B7, the current Stock and exercise prices in B5 and B6, and in B8 the risk free rate  $\ln(1+.05)$ .

**Step 2:** Now use GOAL SEEK (see below) to change the volatility until the predicted BS price for a put matches **actual** price.



We change volatility (B10) until predicted put price (D14) equals \$5.25 (actual put price). From Figure 4.1 we obtain a 53.27% annual volatility. Interestingly enough, in Section 3 historical data indicated 5.7% volatility.

### Remark

If you want EXCEL to figure out the duration of the option enter the current date and expiration dates with the DATEVALUE function and subtract them. To do this we entered in cell G5 the formula =DATEVALUE("6/30/98") and in cell G6 the formula =DATEVALUE("11/22/98"). See Figure 4.2. Note that the DATEVALUE function yields the number of days the date is after January 1, 1900.

**Figure 4.2**

	E	F	G
4			
5	today	6/30/98	35976
6	expire	11/22/98	36121
7			145.00

### ***Using Simulation to Price the Put***

Now we will use Monte Carlo Simulation to price the put. We begin by entering relevant parameters in cells B2:B4 of sheet Dell Sim in file Option.xls. Recall from Section 2 that a fair price for the put is the **expected discounted value of the put's cash flows in a risk neutral world**. In a risk neutral world, the stock will grow at the risk free rate. Therefore we will use (2.1) with  $\mu$  = risk free rate to price the put. We have used the following range names:

- $r_$  = the risk free rate
- $p$  = The current stock price .
- $v$  = volatility
- $d$  = duration
- $x$  = exercise price

We proceed as follows:

**Figure 4.3**

	A	B	C	D	E
1	<b>Pricing Put by Simulation</b>				
2	Current stock price	\$ 94.00		95% CI	
3	Risk free rate	0.053541		Lower Limit	5.090718
4	Duration	0.39726		Upper Limit	5.469088
5	volatility	53.265%			
6	Exercise price	\$ 80.00			
7	Stock price at expiration	91.93506			Name
8	Put cash flows at expiration	0			Description
9	Discounted value of put cash flows	0			Cell

**Step 1:** In cell B7 we use (2.1) to generate DELL's price at the expiration date with the formula

$$=p*EXP((r_-0.5*v^2)*d+RiskNormal(0,1)*v*SQRT(d)).$$

**Step 2:** In cell B8 we compute the cash flows from the put. Recall a put pays nothing if DELL's price on the expiration date exceeds \$80; otherwise the put pays \$80 - (DELL price at expiration).

$$=IF(B7>x,0,x-B7).$$

**Step 3:** In cell B9 we compute the expected discounted value of the put's cash flows with the formula

$$=EXP(-r_*d)*B8.$$

**Step 4:** We select B9 as our output cell and ran 10,000 iterations. From Figure 4.4

**Figure 4.4**

	D	E
2	95% CI	
3	Lower Limit	5.090718
4	Upper Limit	5.469088

	E	F
9	Cell	B9
10	Minimum =	0
11	Maximum =	57.99311
12	Mean =	5.279903
13	Std Deviat	9.459261
14	Variance =	89.47761
15	Skewness	1.900912
16	Kurtosis =	5.931631
17	Errors Calc	0
18	Mode =	0
19	5% Perc =	0
20	10% Perc =	0
21	15% Perc =	0
22	20% Perc =	0
23	25% Perc =	0
24	30% Perc =	0
25	35% Perc =	0
26	40% Perc =	0
27	45% Perc =	0
28	50% Perc =	0
29	55% Perc =	0
30	60% Perc =	0
31	65% Perc =	0.246574
32	70% Perc =	3.815132
33	75% Perc =	7.46743
34	80% Perc =	11.33537
35	85% Perc =	15.56934
36	90% Perc =	20.52942
37	95% Perc =	27.15265

Our best estimate of the put price is \$5.28. We are 95% sure that the put price is between \$5.09 and \$5.47 (see cells E3 and E4). After 10,000 iterations why have we not come closer to the price of the put? The reason is that this put only pays off on very **extreme results** (DELL's price dropping a lot). It takes many iterations to accurately represent extreme values of a Lognormal random variable.

**Remark**

If the stock pays dividends at a rate  $q\%$  per year, then in a risk neutral world the stock price must grow at a rate  $r - q$ . Therefore, for stocks that pay dividends at a rate of  $q\%$  per year, we should price their options by using (2.1) with  $r - q$  replacing  $r$ .

## Section 5-Using The Risk-Neutral Approach to Value Real Options

The risk-neutral approach discussed in Section 2 is very powerful. It enables us to value many quantities which **derive** their value from an underlying asset. In most books, you do this by finding a combination of puts and calls which replicates the payoffs of what you are trying to value. We do not need to do this. Just use the Lognormal to simulate the future value of the underlying asset (growing at risk-free rate) and then make your OUTPUT CELL be the discounted value of the payoffs you receive. The mean of your OUTPUT CELL is the value you seek. Here are six examples:

### Example 5.1

The current price of IBM is 145 and 1/8. In 64 days Gerstner will be paid as follows: For every \$1 increase in stock price up to \$10 he receives \$1 million; For every \$1 increase in stock price over \$10 Gerstner receives \$500,000. What is a fair market value for Lou's option?


**Solution:** We need the volatility of the price of IBM stock. By looking at traded options we can estimate the implied volatility (see file IBMvols.xls and Figure 5.1).

Figure 5.1

	A	B	C	D
1	<b>IBM OPTIONS</b>	April	April	April
2	Current Price	145.125	145.125	145.125
3	Duration	0.175342	0.175342	0.175342
4	Volatility	0.327147	0.327147	0.327147
5	Exercise Price	140	150	160
6	Risk-free Rate	0.0636	0.0636	0.0636
7	d1	0.412352	-0.09128	-0.5624
8	d2	0.275362	-0.22827	-0.69939
9	N(d1)	0.659959	0.463633	0.28692
10	N(d2)	0.608481	0.409717	0.242153
11	Option Price(theory)	11.53395	6.50882	3.324504
12				
13	Option Price(actual)	12	6.25	3.125
14	Squared Error	0.217206	0.066988	0.039802
15				
16	SSE	0.323996		

In row 13 we have listed the price of three call options which expire in 64 days (exercise prices \$140, \$150, and \$160). We have created a Black-Scholes template that, upon entering volatility in row 4 computes a Black-Scholes price in Row 11. We want to find a single volatility which best predicts these prices. We choose B4 to be our changing cell for volatility and minimize sum of squared errors (actual-predicted price)<sup>2</sup> for the three options (cell B16). Our SOLVER WINDOW is as follows:

**Figure 5.2**



We find an implied volatility of 32.7%. We can now value our option. (See Figure 5.3 and file Gerstner.xls).

**Figure 5.3**

	A	B	C	D	E	F
3						
4	Current Price	145.125				
5	Volatility	0.327				
6	Riskfree rate	0.0636				
7	Duration	0.175342			Name	Discounted
8	Normal	0.16111			Description	Output
9	Later Stock Price	148.626			Cell	B11
10	Payoff(millions)	3.500993			Minimum =	0
11	Discounted payo	3.462168			Maximum =	47.52227
12					Mean =	6.374745
13					Std Deviat	8.484468
14					Variance =	71.98619
15					Skewness	1.281692
16					Kurtosis =	4.098358

**Step 1:** In B4-B7 we enter relevant parameters, including the implied volatility.

**Step 2:** In B8 we enter a Normal (0,1) variable with formula

$=RISKNORMAL(0,1)$ .

**Step 3:** In B9 we compute the stock price in 64 days using lognormal growth at the risk-free rate and known volatility:

$$=B4*EXP((B6-0.5*B5^2)*B7+B8*SQRT(B7)*B5)$$

**Step 4:** In B10 we compute Gerstner's payoff with the statement

$$=IF(AND((B9-B4)>=0,(B9-B4)<10),(B9-B4),IF(B9-B4>=10,10+0.5*(B9-(B4+10)),0)).$$

This pays Lou \$1 million per dollar increase up to \$10 and \$.5 million per dollar increase beyond that point.

**Step 5:** In B11 we discount this payoff back to the current time with the formula

$$=EXP(-B6*B7)*B10.$$

**Step 6:** After making B11 our OUTPUT CELL we find from Figure 5.4 that our best estimate is that the fair market value is \$6.37 million.

**Figure 5.4**

	E	F
6		
7	Name	Discounted
8	Description	Output
9	Cell	B11
10	Minimum =	0
11	Maximum =	47.52227
12	Mean =	6.374745
13	Std Deviat	8.484468
14	Variance =	71.98619
15	Skewness	1.281692
16	Kurtosis =	4.098358

### **Example 5.2 An Abandonment Option**

An asset is currently worth \$53,000 and has an annual volatility of 28%. The risk-free rate is 5%. A year from now I may sell the asset for a salvage value of \$500,000. How much is this abandonment option worth?

**Solution:** Let  $V$  = value of asset one year from now. Then a year from now option returns

$$\text{Max}(0, \$500,000 - V) \quad (5.1)$$



If we let asset grow for one year at risk-free rate and with given volatility and take discounted value of (5.1) as our OUTPUT CELL, then the mean of our OUTPUT cell is the value of the abandonment option. Our work is in Figure 5.1 and file Abandonment.xls.

**Figure 5.1**

	A	B	C	D	E	F	G
1	<b>Abandonment</b>						
2	Abandonment value		\$ 500,000.00				
3	r	Volatility	Current Price	Normal(0,1	Price 1 year from now	Value of Option	Discounted value
4	0.05	0.28	\$ 553,000.00	0.387636	\$ 623,093.16	0	0

We proceed as follows:

**Step 1:** In D4 enter a standard normal (mean 0, sigma of 1) random variable with the formula

$$=RISKNORMAL(0,1).$$

**Step 2:** In E4 generate the value of the asset a year from now using the lognormal random variable. The formula is

$$=C4*EXP((A4-0.5*B4^2)+B4*D4).$$

**Step 3:** In F4 we exploit (5.1) to generate the value of the option's cash flows one year from now with the formula

$$=IF(E4<C2,C2-E4,0).$$

**Step 4:** In cell G4 compute the discounted value of the option's cash flows with the formula

$$=EXP(-A4)*F4.$$

**Step 5:** After choosing G4 as our OUTPUT CELL we find the option is worth \$34,093.

**Figure 5.6**

	D	E
7	Name	Discounted val
8	Description	Output
9	Cell	G4
10	Minimum =	0
11	Maximum =	280536.9
12	Mean =	34,092.91
13	Std Deviat	54,870.25
14	Variance =	3.01E+09

***Example 5.3-An Option to postpone (Based on Trigeorgis (1995))***

The current risk-free rate is 8%. We can build a plant now costing \$104 million and gain revenues worth (risk adjusted) \$100 million. Revenues begin one year from now. Therefore current value of project is -\$4 million and project does not appear worthwhile. Suppose, however, project's value has a 60% annual volatility and we can wait one year before investing in this project. What is the worth of this option? Assume that construction costs grow at the risk-free rate.

**Solution**

Our work is in Figure 5.7 and the file Postpone.xls. Let  $V$  = value of project one year from now. Then value of option to postpone is

$$= \max(0, V - e^{.08}(104)) \quad (5.2)$$

This is because we will invest one year from now only if the value of the project exceeds the cost (which one year from now will be  $e^{.08}(104)$ ).

**Figure 5.7**

	A	B	C	D	E	F	G	H	I
1	<b>Postpone Investment</b>								
2									
3									
4	r	Current value	Investment cost	Value with no option	Volatility	Normal (0,1)	Value One year from now	Cash flows in one year from waiting	Discounted value of cash flows
5	0.08	100	104	-4	0.6	1.6431	242.5075	129.8456	119.86261
6									

We now proceed as follows:

**Step 1:** Enter needed parameters in A5-E5.

**Step 2:** In F5 enter a Normal(0,1) with the formula

$=RISKNORMAL(0,1)$ .

**Step 3:** In G5 use the lognormal to compute the value of the asset one year from now with the formula

$=B5*EXP((A5-0.5*E5^2)+F5*E5)$ .

**Step 4:** In H5 we use (5.2) to determine the cash flows obtained in one year with the formula

$=MAX(0,G5-C5*EXP(A5))$ .

**Step 5:** Discount the value of the cash flows back to time 0 in cell I5 with the formula

$=EXP(-A5)*H5$ .

**Step 6:** Make I5 an Output Cell and obtain the following results.

**Figure 5.8**

	D	E
12	Minimum =	0
13	Maximum =	664.869
14	Mean =	22.1453
15	Std Deviation =	50.32
16	Variance =	2532.11
17	Skewness =	4.17166
18	Kurtosis =	31.0005
19	Errors Calculated =	0

Thus the option to postpone improves our position by  $22.14 - (-4) = \$26.14$  million relative to our position if we did project without having option to postpone project.

**Valuing the Option to Expand**

We now modify Example 5.3 to show how we evaluate an option to expand a project.

**Example 5.4**

Assume you have the option to spend \$40 million one year from now on a plant expansion that will increase the project's value by 50%. Value this expansion option.

Solution

Our work is in the file Expand.xls. See Figure 5.9

**Figure 5.9**

	A	B	C	D	E	F
1	<b>Expand for 40 million by 50%?</b>					
2						
3						
4	r	Current value	Investment cost	Value with no option	Volatility	Normal (0,1)
5	0.08	100	104	-4	0.6	-0.16196

	G	H	I	J	K
3					
4	Value One year from now	Cash Flows in One Year	Discounted value of cash flows	Expansion cost	Expansion factor
5	82.10484	83.15726	-27.236177	40	0.5

We proceed as follows;

**Step 1 :** In A5:E5 and J5:K5 we enter relevant parameters for the problem.

**Step 2:** In cell F5 we enter

$=RISKNORMAL(0,1)$ .

This will help us generate the value of the project one-year from now.

**Step 3:** In cell G5 we generate the (random) value of the project one-year from now with the formula

$=B5*EXP((A5-0.5*E5^2)+F5*E5)$ .

**Step 4:** Note that if we choose to expand, our cash flows in one year will equal

$1.5*(\text{value in one year}) - 40$ .

If we do not expand, our cash flows in one year will simply equal the value of the project. Therefore in H5 we compute our cash flows in one year with the formula

$=MAX((1+K5)*G5-J5,G5)$ .

**Step 5:** In cell I5 we compute the discounted value of our cash flows with the formula

$=EXP(-A5)*H5-C5$ .

**Step 6:** We now select cell I5 as our output cell. After running 900 iterations Figure 5.10 indicates that with the option to expand our situation is worth an average of \$14.08 million. Thus the option to expand improves our position over doing the project right away by \$18.08 million.

**Figure 5.10**

	D	E
8		
9	Name	Discounted
10	Description	Output
11	Cell	I5
12	Minimum =	-91.8832
13	Maximum =	987.4983
14	Mean =	14.08345
15	Std Deviat	95.02
16	Variance =	9029.213
17	Skewness	2.609704
18	Kurtosis =	15.51571
19	Errors Calc	0
20	Mode =	-53.04
21	5% Perc =	-72.91
22	10% Perc =	-65.3444
23	15% Perc =	-59.1978
24	20% Perc =	-53.6555
25	25% Perc =	-48.325
26	30% Perc =	-43.0211
27	35% Perc =	-37.731
28	40% Perc =	-32.2731
29	45% Perc =	-24.7762
30	50% Perc =	-15.7222
31	55% Perc =	-5.88407
32	60% Perc =	4.850531
33	65% Perc =	16.87049
34	70% Perc =	30.53909
35	75% Perc =	46.76334
36	80% Perc =	66.62992
37	85% Perc =	92.33498
38	90% Perc =	128.8684
39	95% Perc =	194.7092
40	Filter Minimum =	

***Valuing the Option to Contract***

Now we modify Example 5.3 to value the option to contract a project.

***Example 5.5***

Instead of paying the entire \$104 million plant cost now you must only pay \$50 million now. A year from now you may pay the remaining \$54 million cost (with interest) and obtain the full project value or you may contract the scale of the project by paying only \$25 million. If you contract the scale of the project the project will be worth only 50% of what it would have actually been worth. Value this contraction option.

## Solution

Our work is in the file Contract.xls. See Figure 5.11.

**Figure 5.11**

	A	B	C	D	E	F	G	H
1	<b>Contract Project?</b>							
2	Pay 50 million now							
3	One year from now pay 25 million and get .5 of value or pay $54 \cdot \exp(.08)$ and get whole project							
4	r	Current value	Investment cost now	Volatility	Normal (0,1)	Value One year from now	Cash Flows in One Year	Discounted value of cash flows
5	0.08	100	50	0.6	0	90.48374	31.98624	-20.472979

	I	J	K
3			
4	Cost if Contracted	Cost in one year to keep going(today's \$s)	Contraction Factor
5	25	54	0.5

We proceed as follows:

**Step 1:** Enter the relevant parameters in cells A5:D5 and I5:K5.

**Step 2:** In cell E5 enter

$=RISKNORMAL(0,1)$ .

This random variable will be used in F5 to generate the value of the project in one year.

**Step 3:** In cell F5 we generate the value of the project in one year with the formula

$=B5 \cdot \text{EXP}((A5 - 0.5 \cdot D5^2) + E5 \cdot D5)$ .

**Step 4:** In G5 we compute the cash flows from the project in one year. Note that if we contract the project our cash flows in one year are given by

$.5 \cdot (\text{Value of project}) - \$25 \text{ million}$ .

If we do not contract the project our cash flows in one year are given by

$(\text{Value of Project}) - e^{.04} \cdot (\$54 \text{ million})$

Since we can choose the better of these options in G5 we compute the cash flows from the project in one year with the formula

$$=MAX(F5-EXP(A5)*J5,K5*F5-I5).$$

**Step 5:** In cell H5 we compute the total discounted value of the cash flows from the project with the formula

$$=EXP(-A5)*G5-C5.$$

**Step 6:** Choosing cell H5 as an output cell yields the @RISK output in Figure 5.12.

**Figure 5.12**

	D	E
9		
10	Name	Discounted
11	Description	Output
12	Cell	H5
13	Minimum =	-68.057
14	Maximum =	515.879
15	Mean =	-1.28
16	Std Deviat	62.90
17	Variance =	3956.415
18	Skewness	2.348511
19	Kurtosis =	11.81437

Our simulation indicates that our expected discounted cash flows with the option to contract are \$-1.28 million. This improves our situation over immediately doing project by \$2.72 million.

### ***A 'Pioneer" Option***

Often companies enter into small projects that have a negative NPV. The reason for this is that participation in the smaller project gives the company the option to later participate in a larger project that may have a large positive NPV. Merck Pharmaceuticals, led by their CFO Judy Lewent (see Nichols (1994)) has "pioneered" the use of real option theory. Here is an example of this idea, again based on Trigeorgis (1996).

### ***Example 5.6***

Merck is debating whether to invest in a pioneer biotech project. They estimate the worth of this project to be -\$56 million. Investing in the pioneer project gives Merck the option to own, if they desire, a much bigger technology that will be available in 4 years. If Merck does not participate in the Pioneer project **they cannot own the bigger project.** The big project will require 1.5 billion in cash 4 years from now. Currently Merck estimates the NPV of the cash flows from the bigger project to be \$57 million. What should Merck do? Risk free rate is 10% and annual volatility of big project is 35%.



## Solution

Our solution is in the file Pioneer.xls. See Figure 5.13.

**Figure 5.13**

	A	B
1	<b>Buying a Pioneer Project</b>	
2		
3	NPV of pioneer project	-5.6E+07
4	Current NPV of new technology	5.97E+08
5	Year 4 cost of New Technology	1.5E+09
6	r	0.1
7	Annual volatility	0.35
8	Normal (0,1)	0.513126
9	Value of project in 4 years	9.98E+08
10	NPV in year 4 of doing new Project	0
11	Discounted NPV of doing new project	0

We proceed as follows:

**Step 1:** Enter relevant parameter values in B3:B7.

**Step 2:** In B8 generate a standard normal that will be used to generate the value of the project in four years with the formula

$$=RISKNORMAL(0,1).$$

**Step 3:** In cell B9 generate the value of the project in four years with the formula

$$=B4*EXP((B6-0.5*B7^2)*4+B8*SQRT(4)*B7).$$

**Step 4:** In cell B10 we compute the NPV (in Year 4 \$s) of doing the new Project

$$=MAX(B9-B5,0).$$

This assumes, of course, that we only do the new project if it is worthwhile.

**Step 5:** In cell B11 we compute the value (in today's dollars) of doing the new project.

$$=EXP(-4*B6)*B10.$$

**Step 6:** We choose B11 as our Output cell. Our @RISK results (900 iterations) are in Figure 5.14.

**Figure 5.14**

	D	E
4	Name	Discounted NPV of doing
5	Description	Output
6	Cell	<b>B11</b>
7	Minimum =	0
8	Maximum =	4.25E+09
9	Mean =	69,091,440.00
10	Std Deviat	274,982,500
11	Variance =	7.56E+16
12	Skewness	6.705864

It appears that the value of the option to invest in the new project is \$69 million. This more than outweighs the negative NPV of the Pioneer project, so we should go ahead with the Pioneer project.

### Remarks

Chapter 21 of Brealey and Myers (1996) contains an excellent introduction to real options. The two books by Trigeorgis contain a comprehensive (though advanced) discussion of real options. We refer the reader to Luenberger (1997) for an excellent discussion of real options.

Of course, it is difficult to come up with volatility for a project. Looking at implied volatilities for companies in a similar industry may be helpful. Probably the best strategy is to use the =RISKSIMTABLE to value the option for a wide range of volatilities.

## Chapter 6-Pricing an American Option with Binomial Trees

We cannot use @RISK to price an American option because to determine the cash flows from the option we need to account for the possibility of early exercise. American options are usually priced with **binomial trees**. We divide the duration of the option into smaller time periods (usually weeks or months). During each time period the stock price either increases by a factor  $u$  or decreases by a factor  $d$ . We assume  $d = 1/u$ . Let  $\Delta t$  equal length of period in tree. The probability of an increase each period ( $p$ ) is chosen in conjunction with  $u$  and  $d$  so the stock price grows on average at the risk-free rate  $r$  and has an annual volatility of  $\sigma$ . We let  $q = 1 - p$  be the probability of a decrease during each period. To ensure that our tree matches the mean growth rate and volatility per unit time of an asset following a Lognormal random variable with parameters  $r$  and  $\sigma$  we must choose (see Chapter 15 of Hull (1997))

$$u = e^{\sigma \sqrt{\Delta t}}$$

$$d = 1/u$$

$$a = e^{r \cdot \text{deltat}}$$

$$q = 1 - p$$

$$p = \frac{a - d}{u - d}$$

We set up our tree in file American.xls. Let's price a 5-month American put with

- ◆ Current stock price = \$50
- ◆ Exercise price = \$50
- ◆ Risk-free rate = 10%
- ◆ Annual volatility = 40%
- ◆ Deltat = 1 month = .083 years.

All this data plus the previous definitions of  $u$ ,  $d$ ,  $a$ ,  $q$ , and  $p$  are input in the range A1:B13 of the spreadsheet. We have used range names to make the tree easier to explain. Our work is in the sheet pricing of file American.xls.

### ***The Stock Price Tree***

We begin by determining the possible stock prices during the next 5 months. Column B has today's (Month 0) price, Column C Month 1, etc.

**Step 1:** Enter today's stock price (\$50) in B16 with the formula

=B3.

**Step 2:** By copying the formula

=u\*B16

from C16 to D16:G16 we obtain the price each month when there have been no down moves.

**Step 3:** To compute all other prices note that in each column as we move down a row the price is multiplied by a factor ( $d/u$ ). Also note that for Month  $I$  there are  $I + 1$  possible prices. This allows us to compute all prices by copying from C17 to C17:G21 the formula

=IF(\$A17<=C\$15,(d/u)\*C16,"-").

As we move down each column, the prices are successively multiplied by  $d/u$ . Also, our formula places a "-" where a price does not exist.

### ***The Optimal Decision Strategy***

We now work backwards to find the value of the American Put. Remember at each node the value of the put equals the **expected discounted value of future cash flows from the put**.

**Step 1:** At month 5 the option is just worth maximum (exercise price – current stock price). Thus we enter

$=MAX(0, \$B\$4-G16)$

in cell G24 and copy this formula down to G29.

**Step 4:** In Month 4 (and all previous months) the value of the option at any node is

Maximum(value from exercising now,  $(1/(1+(.1/12))) * (p * (\text{value of option for up move}) + q * (\text{value of option for down move}))$ ).

For example, in F28 the value of option is

$\max(50 - 31.50, (1/1.0083) * (.507 * (\$14.64) + .493 * (\$21.93))) = \$18.50$ .


Since this maximum is attained by exercising now, if this node occurs we would exercise the option now. At the node in F26 the maximum is attained by not exercising. To implement this decision-making procedure we enter in cell F24 the formula

$=IF(\$A24 <= F\$23, MAX(\$B\$4-F16, (1/(1+r_*deltat)) * (p * G24 + (1-p) * G25)), "-")$ .

Copying this formula to B24:F29 generates value of option for all possible prices during Months 1-4 and places a "-" in any cell where there is no actual stock price. In cell B24 we find the estimated value of the put \$4.49. Of course, \$4.49 is an approximation to the put value. As *deltat* grows small, however, our price will converge to the actual price of the put if the stock grew according to a Lognormal random variable.

### ***Using Conditional Formatting to Describe the Optimal Exercise Policy***

Assuming the stock grows at the risk free rate, how would we react to actual price changes? Suppose the first three months have down moves. We do not exercise during first two months but we exercise after third down move. Suppose first four months are down, up, down, down. Then we exercise after the fourth month. To make this clearer, we use EXCEL'S CONDITIONAL FORMAT OPTION to format the spreadsheet so that cells for which option would be exercised are in bold face. We begin by noting that **it is optimal to exercise the option at a Month and price if and only if the value of the cell corresponding to the month and price equals (exercise price) - (stock price)**. To indicate the cells where exercising the option is optimal begin in cell G24 by selecting FORMAT CONDITIONAL FORMAT and enter the following:



Then click on **FORMAT BOLD**. This dialog box ensures that if the option is exercised in the state with 5 up moves in period 5 then a bold font is used. The interpretation of

$$=($B$4-G16)=G24$$

is the format takes hold only if  $(\$B\$4-G16)=G24$  which is equivalent to option being exercised. If we **EDIT COPY** cell G24 and **PASTE SPECIAL FORMATS** to the range B24:G29 we ensure that any period and number of up moves for which exercise of the option is optimal will be indicated in bold font.

**Figure 6.1**

	A	B	C	D	E	F	G
1	<b>American Option</b>						
2	<b>Put</b>						
3	Current price	\$ 50.00					
4	Exercise price	\$ 50.00					
5	r	0.1					
6	sigma	0.4					
7	t	0.416667					
8	deltat	0.083333					
9	u	1.122401					
10	d	0.890947					
11	a	1.008368					
12	p	0.507319					
13	q	0.492681					
14		<b>Time</b>					
15	<b>Stock Prices</b>	0	1	2	3	4	5
16	0	\$ 50.00	\$ 56.12	\$ 62.99	\$ 70.70	\$ 79.35	\$ 89.07
17	1		\$ 44.55	\$ 50.00	\$ 56.12	\$ 62.99	\$ 70.70
18	2		-	\$ 39.69	\$ 44.55	\$ 50.00	\$ 56.12
19	3		-	-	\$ 35.36	\$ 39.69	\$ 44.55
20	4		-	-	-	\$ 31.50	\$ 35.36
21	5		-	-	-	-	\$ 28.07
22	<b>Down Moves</b>						
23	<b>Put Value</b>	0	1	2	3	4	5
24	0	\$ 4.49	\$ 2.16	\$ 0.64	\$ -	\$ -	\$ -
25	1	-	\$ 6.96	\$ 3.77	\$ 1.30	\$ -	\$ -
26	2	-	-	\$ 10.36	\$ 6.38	\$ 2.66	\$ -
27	3	-	-	-	<b>\$ 14.64</b>	<b>\$ 10.31</b>	<b>\$ 5.45</b>
28	4	-	-	-	-	<b>\$ 18.50</b>	<b>\$ 14.64</b>
29	5	-	-	-	-	-	<b>\$ 21.93</b>

**Sensitivity Analysis**

Using one-way data tables it is easy to see how changes in various input parameters change the price of the put. We varied the annual volatility of the stock from 10%-70%. Figure 6.2 (a one-way data table with Input Cell sigma) shows how an increase in volatility greatly increases the value of the put. Increased volatility gives us a larger chance of a big price drop, which increases the value of the put.

**Figure 6.2**

	H	I
7	<b>Volatility</b>	4.489053
8	0.1	0.680146
9	0.2	1.91777
10	0.3	3.197364
11	0.4	4.489053
12	0.5	5.799995
13	0.6	7.104648
14	0.7	8.400682

Figure 6.3 shows how a change in the exercise price of the stock changes the value of the put. We used a one-way data table with input cell B4.

**Figure 6.3**

	K	L
7	<b>Ex Price</b>	4.489053
8	\$ 45.00	2.139609
9	\$ 46.00	2.609498
10	\$ 47.00	3.079387
11	\$ 48.00	3.549275
12	\$ 49.00	4.019164
13	\$ 50.00	4.489053
14	\$ 51.00	4.980604
15	\$ 52.00	5.484364
16	\$ 53.00	5.988124
17	\$ 54.00	6.52469
18	\$ 55.00	7.092245
19	\$ 56.00	7.733071

As the exercise price increases, the value of the put increases because an increased exercise price increases the number of values for which the put is "in the money".

Finally, Figure 6.4 shows that increases in the risk-free rate decrease the value of the put.

**Figure 6.4**

	N	O
7	<b>r</b>	4.489053
8	0.01	5.300677
9	0.02	5.207424
10	0.03	5.114923
11	0.04	5.023183
12	0.05	4.932208
13	0.06	4.842007
14	0.07	4.752584
15	0.08	4.663947
16	0.09	4.576102
17	0.1	4.489053
18	0.11	4.405002
19	0.12	4.333566

This is because an increase in the risk-free rate makes the payoff from the put (which occurs in the future) less valuable.

### ***Relationship to an Abandonment Option***

Often an option to abandon a project may be thought of as a put. To see this in the current context, suppose the current value of a project is \$50 million, the risk-free rate is 10%, and the project has a 40% annual volatility. Any time in the next five months we may abandon the project and receive \$50 million. To determine the value of this

abandonment option we would proceed exactly as we proceeded to value the put. We would have found the value of the abandonment option to be \$4.49 million.

### Computing the Early Exercise Boundary

Given the price of the stock today is \$50 it would be nice to know, in advance what we would do (exercise or not exercise) for any given price during a future period. For example, would we exercise if Month 1 price were \$42? If we have not exercised during first three months would we exercise during Month 4 if price were \$40? Answering this question requires that we compute the **early exercise boundary** for each period. It turns out for each month there exists a "boundary price"  $p(t)$  such that we will exercise during Month  $t$  (assuming option has not been exercised) if and only if Month  $t$  price is less than or equal to  $p(t)$ . Together  $p(1)$ ,  $p(2)$ ,  $p(3)$ ,  $p(4)$ , and  $p(5)$  define the early exercise boundary for the put. To find the early exercise boundary it is convenient to make four copies of our original sheet. To copy a sheet put the cursor on the sheet name, hold down the left mouse button, and drag the sheet to another tab. We have renamed our four copies Ex Bound 1, Ex Bound 2, etc. In sheet Ex Bound 1 we determine  $p(1)$  as follows. The value  $p(1)$  for which we exercise during Month 1 if and only if  $p \leq p(1)$  can be found by observing that  **$p(1)$  is the largest Month 1 price for which Exercise price -  $p(1)$  equals the Month 1 value of the option.** To find  $p(1)$  (see Figure 6.5) we proceed as follows:

Figure 6.5

	A	B	C	D	E	F	G
1	<b>American Option</b>						
2	<b>Put</b>						
3	Current price	\$ 50.00					
4	Exercise price	\$ 50.00					
5	r	0.1					
6	sigma	0.4					
7	t	0.416667					
8	deltat	0.083333					
9	u	1.122401					
10	d	0.890947					
11	a	1.008368					
12	p	0.507319					
13	q	0.492681					
14		Time					
15	<b>down moves</b>	0	1	2	3	4	5
16		0 \$ 50.00	\$ 39.19	\$ 43.98	\$ 49.37	\$ 55.41	\$ 62.19
17		1	\$ 31.11	\$ 34.91	\$ 39.19	\$ 43.98	\$ 49.37
18		2	-	\$ 27.71	\$ 31.11	\$ 34.91	\$ 39.19
19		3	-	-	\$ 24.69	\$ 27.71	\$ 31.11
20		4	-	-	-	\$ 22.00	\$ 24.69
21		5	-	-	-	-	\$ 19.60
22		ex-price	\$ 10.81				
23	<b>Put Value</b>	0	1	2	3	4	5
24		0 14.6728	<b>10.81387</b>	6.84145	3.096029	0.309755	0.00
25		1	<b>18.89456</b>	<b>15.08723</b>	<b>10.81387</b>	<b>6.017455</b>	<b>0.63</b>
26		2	-	<b>22.28669</b>	<b>18.89456</b>	<b>15.08723</b>	<b>10.81</b>
27		3	-	-	<b>25.30891</b>	<b>22.28669</b>	<b>\$ 18.89</b>
28		4	-	-	-	<b>28.00154</b>	<b>\$ 25.31</b>
29		5	-	-	-	-	<b>\$ 30.40</b>




**Step 1:** In cell C16 insert a trial value for  $p(1)$ . Note that the way we have set up the price tree ensures that the prices in B16 and C17 have no effect on the value of the put computed in C24.

**Step 2:** Assuming that the Month 1 price equals the value in C16, we compute in C22 the value if the put is exercised in Month 1 with the formula

$$=B4-C16.$$

**Step 3:** We can now use SOLVER to determine  $p(1)$ . To find  $p(1)$  note that  $p(1)$  is the **largest** price (entered in C16) for which the value of exercising now (in cell C22) equals the Month 1 value of the option (computed in C24). Therefore the following SOLVER WINDOW let's us compute  $p(1)$ .



We find that  $p(1) = \$39.19$ . Thus if the Month 1 price is below  $\$39.19$  we should exercise, otherwise go on. The reader should try a variety of Month 1 prices in B16 to convince herself that it is optimal to exercise for any price below  $\$39.19$  and continue for any price above  $\$39.19$ . Of course, we are assuming that exercise can only occur at time  $=1, 2, \dots, 5$ , but it turns out that even if many more points of exercise were allowed,  $p(1)$  would be fairly close to  $\$39.19$ . In a similar fashion we find the rest of the early exercise boundary to be the following:

Time	Exercise if price <=
1	\$39.19
2	\$39.41
3	\$41.64
4	\$45.28
5	\$50.00

### **Simulating the Actual Cash Flows from an American Option**

Now that we have computed the early exercise boundary, it is easy to simulate the actual cash flows generated by optimal exercise of the put. Our work is in file VAR.xls and Figure 6.6. We assume that the actual stock price follows a Lognormal random variable with  $\mu = .20$ .

**Figure 6.6**

	A	B	C	D	E	F	G
1	<b>American Option</b>						
2	<b>Put</b>						
3	Current price	\$ 50.00					
4	Exercise price	\$ 50.00					
5	r	0.1					
6	sigma	0.4					
7	t	0.416667					
8	deltat	0.083333					
9	u	1.122401					
10	d	0.890947					
11	a	1.008368					
12	p	0.507319					
13	q	0.492681					
14	growth rate	0.2	1	2	3	4	\$ 5.00
15	Ex Bound		39.19	39.41	41.64	45.28	50
16	Price	50	47.87002	37.79165	43.94178	51.86517	52.0111
17	Value if Exercised		0	12.20835	0	0	0
18	Value to date		0	12.00656	12.00656	12.00656	12.00656

**Step 1:** In cells C16:G16 generate a 5-month stock price path by copying from C16 to D16:G16 the formula

$$=B16*EXP((\$B$14-0.5*sigma^2)*(1/12)+RiskNormal(0,1)*sigma*SQRT(1/12)).$$

**Step 2:** Enter the exercise boundary in C15:G15.

**Step 3:** In C17:G17 we compute the cash flow from the put, assuming it has not yet been exercised by copying from C17 to D17:G17 the formula

$=IF(C16 < C15, \$B\$4 - C16, 0)$ .

These formulas ensure that if the current price is less than the exercise boundary we receive exercise price - current stock price. Otherwise, we receive nothing.

**Step 4:** In C18:G18 we compute the actual discounted payoff earned from the put. In C18 we compute the Month 1 payoff with the formula

$=C17 * EXP(-r_*C14/12)$ .

To compute the final discounted payoff received we copy from D18 to E18:G18 the formula

$=IF(C18 > 0, C18, EXP(-r_*D14/12) * D17)$ .

If the option has already been exercised, we retain the discounted value already received. Otherwise, we enter the discounted value (possibly 0) earned during the current Month. Then cell G18 gives the **actual discounted cash flow** from the put.

**Step 5:** We ran 1600 iterations with output cell G18 and obtained the output in Figure 6.7.

**Figure 6.7**

	H	I
13	Cell	G18
14	Minimum =	0
15	Maximum =	19.67661
16	Mean =	3.711043
17	Std Deviat	4.89601
18	Variance =	23.97092
19	Skewness	0.94251
20	Kurtosis =	2.479961
21	Errors Calc	0
22	Mode =	0
23	5% Perc =	0
24	10% Perc =	0
25	15% Perc =	0
26	20% Perc =	0
27	25% Perc =	0
28	30% Perc =	0
29	35% Perc =	0
30	40% Perc =	0
31	45% Perc =	0
32	50% Perc =	0.00E+00
33	55% Perc =	0.443471
34	60% Perc =	2.704098
35	65% Perc =	4.871075
36	70% Perc =	6.391032
37	75% Perc =	7.866961
38	80% Perc =	8.973576
39	85% Perc =	10.41997
40	90% Perc =	11.45398
41	95% Perc =	12.9345
42	Value	0
43	Percentage	53.75%

We find an expected discounted cash flow from the put of \$3.71. By using the TARGET command (see I42 and I43) we find there is a 46% that the put will return a positive cash flow.

## **Section 7-Using RISKOPTIMIZER to Model a Gold Mine Lease and Expansion**

In the next two sections we will show how to use RISKOPTIMIZER (an EXCEL add-in) to price some complex American options. RISKOPTIMIZER allows us to choose values for a set of Adjustable cells which optimize a random function of a given target cell. For example, we could choose the number of put options on the market which maximize the Value at RISK (5<sup>th</sup> percentile) of an index fund. As another example, we could choose an asset allocation that would maximize the Sharpe ratio of a retirement

portfolio. To solve such as problem RISKOPTIMIZER runs a simulation for given values of the adjustable cells until the random function being optimized converges. Then another simulation is run for a new set of adjustable cells. Genetic algorithms are used to determine a set of values for the adjustable cells that comes close to optimizing the random function (mean ,standard deviation, 5<sup>th</sup> percentile, etc) of the target cell.

Suppose the uncertainty concerning an investment decision deals primarily with risk involving the price of a commodity. Then the risk neutral approach discussed in Section 2 implies (in our analysis we ignore convenience costs and storage yields) that an investment decision may be valued as the expected discounted (at risk-free rate) value of cash flows generated by the investment under the assumption that the price of the relevant commodity grows at the risk free rate. This is because in a risk neutral world the value of any asset must grow at the risk free rate. In the next two chapters we will use the real options approach to value three options involving a gold mine. Our inspiration comes from Chapter 12 of Luenberger (1997).

### ***Example 7.1***

Assume the current price of gold is \$400 per ounce. The price of gold evolves through time as a Lognormal random variable with an annual volatility of 30%. The current risk-free rate is 10%. It costs \$300 per ounce to extract gold, so we will only extract gold during a year in which the price exceeds \$300. Up to 10,000 ounces of gold can be mined and sold each year. Assuming cash flows occur at the end of each year, value a 10-year lease on this gold mine.

### **Solution**

Our work is in file gold1.xls. See Figure 7.1.

**Figure 7.1**

	A	B	C	D	E	F	G	H	I	J	K
1	r	0.1									
2	sigma	0.3									
3	extraction rate	10000									
4	current price	400									
5	ext cost	300									
6											
7	time	0	1	2	3	4	5	6	7	8	9
8	price	400	422.616	446.511	471.757	498.4307	526.612	556.387	587.85	621.08	656.2
9	mine?	yes	yes	yes	yes	yes	yes	yes	yes	yes	yes
10	profit	1000000	1226162	1465112	1717572	1984307	2266123	2563873	3E+06	3E+06	4E+06
11											
12											
13	npv	1.21E+07			Name	npv / yes					
14					Description	Output					
15					Cell	B13					
16					Minimum	909090.9					
17					Maximum	1.63E+08					
18					Mean =	1.92E+07					

**Step 1:** We enter relevant parameters in B1:B5. We name cell B1 r\_ and name B2 sigma.

**Step 2:** In B8:K8 we use (2.1) to generate gold prices (per ounce) for the next 10 years. **To perform a risk neutral valuation, we assume price of gold grows at risk free rate.** This is because in a risk neutral world, gold prices would grow (ignoring convenience yields and storage costs) at the risk free rate. In B8 we enter the current gold price \$400. In C8:K8 we use (26.1) to generate future gold prices by copying from C8 to D8:K8 the formula

$$=B8*EXP((r_-0.5*sigma^2)+sigma*RiskNormal(0,1)).$$

**Step 3:** In B9:K9 we indicate that we will mine gold if and only if price is greater than \$300 per ounce. Simply copy from B9 to C9:K9 the formula

$$=IF(B8>B5,"yes","no").$$

**Step 4:** In B10:K10 we compute the annual profit from mining. If we do not mine profit is \$0. If we mine we simply take extraction rate times profit earned per ounce extracted. Simply copy from B10 to C10:K10 the formula

$$=IF(B9="yes",B3*(B8-B5),0).$$

**Step 5:** In cell B13 we compute the discounted value (**at risk free rate, because that is proper discount rate in a risk neutral world!**) of our profits with the formula

=NPV(r\_,B10:K10)..

**Step 6:** Running an @RISK simulation we find (see cell F18) that the option to lease the gold mine is worth \$19.2 million.

### Valuing an Expansion Option

Let's assume that we can expand the capacity of the mine. If we expand the capacity of the mine we can extract 50% more gold per year, but extraction cost will increase by \$20 per ounce. It will cost \$8 million to expand the mine. By how much does this expansion option increase the value of the lease?

### Solution

Our work is in Figure 7.2 and file gold2.xls. We need to have adjustable cells for each year that determine whether or not we should expand (assuming we have not already expanded). It can easily be shown that for each year  $t$  there exists a cutoff point  $p(t)$  such that we should expand if year  $t$  price is at least  $p(t)$  and not expand if year  $t$  price is less than  $p(t)$ . We will use RISKOPTIMIZER to find the cutoffs that maximize the expected discounted (at risk free rate) value of cash flows over 10 years. This problem could also be solved using binomial trees (see Section 6).

**Figure 7.2**

	A	B	C	D	E	F	G	K
1		current	expansion					
2	r	0.1						
3	sigma	0.3						
4	extraction rate	10000	15000					
5	current price	400						
6	ext cost	300	320					
7	cost		8.00E+06					
8	time	0	1	2	3	4	5	9
9	price	400	469.83516	421.058	486.485	200.03	131.784	150.222
10	cut for expand?	1070	623	693	724	684	800	1882
11	expanded already?	no	no	no	no	no	no	no
12	expand now?	no	no	no	no	no	no	no
13	profit	1000000	1698351.6	1210581	1864854	0	0	0
14	buildingcost	0	0	0	0	0	0	0
15	profit	1000000	1698351.6	1210581	1864854	0	0	0
16	npv	4.50E+06						
17								
18								
19								
20								

Mean = 2.1638E+007

**Step 1:** Enter parameter values in rows 2-7 and generate gold prices for years 0-9 as before.

**Step 2:** In B10:K10 enter trial cutoff points for the p(t).

**Step 3:** In B11:K11 determine if have already expanded. In B11 enter "no" because we have not yet expanded. In C11:K11 determine if by beginning of year we have already expanded by copying from C11 to D11:K11 the formula

$=IF(OR(B11="yes",B12="yes"), "yes", "no")$ .

If we have already expanded at beginning of previous year or we expand during previous year, this formula enters a "yes", otherwise it enters a "no".

**Step 4:** In B12:K12 we determine if we expand during the current year. We expand if and only if we have not yet expanded and price of gold during current year exceeds cutoff . Simply copy from B12 to C12:K12 the formula

$=IF(AND(B11="no",B9>=B10), "yes", "no")$ .

This formula ensures that we expand if and only if current price exceeds cutoff and we have not yet expanded.

**Step 5:** In B13:K13 we determine profit (excluding building cost) for each year. Note that if we have expanded and current price is at least \$320 per ounce we will sell 15,000,000 ounces at current price and it will cost us \$320 per ounce to produce the gold. If we have not expanded and current price is at least \$300 we will sell 10,000,000 ounces at current price and it will cost us \$300 an ounce to produce the gold. Otherwise, we extract no gold. Copying from B13 to C13:K13 the formula

$=IF(OR(B11="yes",B12="yes"),MAX(0,$C$4*(B9-$C$6)),MAX(0,$B$4*(B9-$B$6)))$

will determine the correct profit for each year.

**Step 6:** In B14:K14 we compute the expansion cost (if any) incurred during the current year. If we expand, a cost of \$8 million is incurred, otherwise no cost is incurred. Copying from B14 to C14:K14 the formula

$=IF(B12="yes", $C$7, 0)$

picks up the expansion cost for each year.

**Step 7:** In B15:K15 we compute profit for each year by subtracting expansion cost from mining profit. Just copy from B15 to C15: K15 the formula


$=B13-B14$ .



**Step 8:** In cell B16 we compute the expected discounted (at risk free rate) value of profits from gold mine with the formula

$$=NPV(r_,B15:K15).$$

**Step 9:** We can now use RISKOPTIMIZER to find out the value of the lease with expansion option. Our Settings box follows:



We try and choose the expansion cutoff points (B10:K10) to maximize the expected discounted NPV (cell B16) of profits. We constrain B10:K10 to be an integer between \$320 and \$2000. From Figure 7.2 we find that RISKOPTIMIZER values this situation at \$21.6 million. Thus the expansion option has increased the mine's value by \$21.6 - \$19.2 = \$2.4 million.

### Remark

The cutoffs  $p(t)$  found by RISKOPTIMIZER may not be the exact cutoffs that maximize profit. If a particular  $p(t)$  does not have much effect on value of option, then RISKOPTIMIZER will not expend much effort in accurately determining  $p(t)$ . We can be fairly sure, however, that the value of the option found by RISKOPTIMIZER is fairly close to the actual maximum expected discounted value that could be obtained by exhaustively searching all combinations of  $p(t)$ 's.

## Section 8-Using RISKOPTIMIZER to Analyze A Startup and Shutdown Option

We now extend the gold mine example of Section 7 to allow the mine owner to shut down the mine (to save fixed operating costs) when the price of gold is low and reopen the mine when the price of gold increases.

### Example 8.1

We are interested in the value of owning a gold mine over the next 20 years. The current price of gold is \$400 per ounce and the annual volatility is 30%. It costs \$250 per ounce to extract an ounce of gold. In addition, a fixed cost of \$1,000,000 is incurred each year the mine is open. After observing the price of gold for the year we may (for a cost of \$1,500,000) shut down open mine or (for a cost of \$2,000,000) open a shut down mine. Use real options theory to value this situation. Assume the mine is open at the beginning of the problem.

### Solution

Our work is in the file goldshut.xls. See Figure 1. As in Section 7, we let gold price grow at the risk-free rate. Then we value the situation as the expected discounted (at risk-free rate) value of our profits given we act each year in a way that maximizes our expected NPV.

Figure 8.1

	A	B	C	D	E	F	G	T	U
1	r	0.1							
2	sigma	0.3							
3	extraction rate	10000							
4	current price	400							
5	ext cost	250							
6	fixed cost	1000000							
7	start up cost	2000000							
8	shutdown cost	1500000							
9									
10	time	0	1	2	3	4	5	18	19
11	price	400	1060.235	884.4431	1098.061	1310.676	1220.669	3832.948	5674.676
12	open at begining?	yes	yes	yes	yes	yes	yes	yes	yes
13	open cutoff	410	1909	502	401	368	385	400	1999
14	do we open?	no	no	no	no	no	no	no	no
15	shut cutoff	181	215	264	255	85	244	261	17
16	do we shut?	no	no	no	no	no	no	no	no
17	status?	yes	yes	yes	yes	yes	yes	yes	yes
18	nonextraction costs	1000000	1000000	1000000	1000000	1000000	1000000	1000000	1000000
19	extraction cash flows	1500000	8102353	6344431	8480608	10606756	9706694	35829479	54246763
20	total cash flows	500000	7102353	5344431	7480608	9606756	8706694	34829479	53246763
21	npv	\$144,442,305.76							
22									
23									
24									
25									

Mean = 4.3479E+007

The key to our model is to have "cutoff" prices of gold for each year which determine when we open a shut mine or close an open mine. In Row 13 we have an Adjustable cell  $O(t)$  for each year. If price of gold during year  $t$  is at least  $O(t)$  we open a shut mine. In Row 15 we have an Adjustable cell  $S(t)$  for each year. If price of gold during year  $t$  is less than or equal to  $S(t)$  we will shut down an open mine. The rest of the spreadsheet is simple bookkeeping. Note that we will mine gold with open mine if and only if per ounce gold price is more than \$250.

**Step 1:** In A1:B8 we enter problem parameters.

**Step 2:** In B11:U11 we use the Lognormal random variable (as described in Section 2) to generate 20 years of random gold prices.

**Step 3:** In B13:U13 we enter trial values of the  $O(t)$ , price cutoffs for opening a shut mine.

**Step 4:** In B15:U15 we enter trial values of the  $S(t)$ , price cutoffs for closing an open mine.

**Step 5:** In Row 12 we will keep track of whether the mine is open at the beginning of the year. See Step 9.

**Step 6:** In B14:U14 we determine if a shut mine was opened during the year. To do this copy the statement

$=IF(AND(B12="no",B11 \geq B13),"yes","no")$

from B14 to C14:U14. This ensures that the mine will be opened during year  $t$  if year  $t$  gold price is at least  $O(t)$  and mine was closed at end of last year.

**Step 7:** In B16:U16 we determine if an open mine is shutdown during the current year. To do this copy the statement

$=IF(AND(B12="yes",B11 \leq B15),"yes","no")$

from B16 to C16:U16

**Step 8:** In B17:U17 we determine the status of the mine at the end of the year by copying from B17 to C17:U17 the formula

$=IF(B14="yes","yes",IF(B16="yes","no",IF(AND(B12="yes",B16="no"),"yes","no")))$ .

If the mine is opened during the current year, this formula generates a "yes". If the mine is closed during the current year, this formula generates a "no". If the mine was open at the beginning of the year and was not shut, this formula generates a "yes". Otherwise the mine is closed and the formula generates a "no".

**Step 9:** In B12:U12 we generate the status of the mine at the beginning of each year. In B12 we enter a "yes" because mine starts open. In C12:U12 we just recopy the status at the end of the previous year by copying from C12 to D12:U12 the formula

$=B17$ .

**Step 10:** In B18:U18 we generate the fixed operating costs, start-up costs, and shut-down costs for each year by copying from B18 to C18:U18 the formula

$=IF(B17="yes",\$B\$6,0)+IF(B14="yes",\$B\$7,0)+IF(B16="yes",\$B\$8,0)$ .

The first =IF statement incurs a fixed cost if mine is operating. The second =IF statement incurs a startup cost if we open the mine. The third =IF statement incurs a shutdown cost if we close the mine.

**Step 11:** In B19:U19 we compute the extraction cash flows for each year. Note that we will mine to capacity if and only if mine is open and current price of gold is more than \$250. Simply copy from B19 to C19:U19 the formula

$=IF(B17="yes",MAX(0,(B11-\$B\$5)*\$B\$3),0)$ .

If we are open we extract our capacity if gold price exceeds \$250 and earn (gold price - 250) per ounce extracted. Otherwise we earn nothing.

**Step 12:** By copying from B20 to C20:U20 the formula


$=B19-B18$

we compute our profit for each year.

**Step 13:** In cell B21 we compute the NPV (at risk-free rate) of our profits with the formula (we named risk-free rate  $r_$ )

$=NPV(r_,B20:U20)$ .

**Step 14:** We are ready to use RISKOPTIMIZER to choose gold price cutoffs that maximize our expected discounted NPV. Our Settings box follows:



We are trying to maximize mean NPV (cell B21). Our adjustable cells are price cutoffs for opening the mine (B13:U13) and price cutoffs for shutting the mine (B15:U15). We constrain the price cutoffs for opening to be between \$0 and \$2000. We constrain the price cutoffs for shutting to be at most \$300. RISKOPTIMIZER reports that the value of owning this mine for 20 years is \$43.5 million.

#### Remark

Again, we have ignored storage costs and convenience yields in our analysis.

## Section 9- Timing Market Entry

In this section we show how RISKOPTIMIZER can be used to model complex sequential decisions in which our choice of later decision influences our earlier decision. Essentially, we illustrate how RISKOPTIMIZER can be used to analyze, within the context of simulation, decision-making under uncertainty problems that involve continuous random variables such as market share and market size. To illustrate the idea, we consider the following important problem: should a high-tech company enter a new market now or wait till later?

When a high-tech company is trying to determine whether to market a new product there is a great deal of uncertainty. Should we enter the market now, not knowing if growth in the market will be high or low? If we enter now, we get a larger market share than we would get by entering later. The downside is we may have a high share of a small market. If we wait a few years, we will know if market growth will be high or low, but we have missed out on several years of sales and will probably obtain a reduced market share due to our late entry. RISKOPTIMIZER can be used to make these complex

entry decisions. The following example explores some of the important issues involved in the decision to enter a market.

**Example 9.1**

Fell computer is trying to determine whether to enter a market for a new type of computer. They can enter the market now or wait two years and enter the market at the beginning of Year 3. It will cost \$3.5 billion to enter this market. Year 1 market size is unknown but is believed to follow a normal random variable with mean 5,000,000 and standard deviation 1,000,000. There are three scenarios for market growth: High, Medium, and Low. Relevant information for each scenario is given below:

Scenario	Unit profit contribution	Market size growth years 2-6	Market size growth after year 6	Probability of scenario
High	\$90	$\mu = 100\%$ $\sigma = 20\%$	$\mu = 15\%$ $\sigma = 20\%$	.1
Medium	\$60	$\mu = 20\%$ $\sigma = 15\%$	$\mu = 5\%$ $\sigma = 15\%$	.5
Low	\$40	$\mu = -5\%$ $\sigma = 15\%$	$\mu = -20\%$ $\sigma = 15\%$	.4

For example, the High scenario is estimated to have a 10% chance of occurring. If the High scenario occurs a unit profit of \$90 can be earned on each sale. Annual growth during Years 2-6 will follow a normal distribution with mean 100% and standard deviation 20%. After Year 6, the product will reach the mature portion of the product life cycle and grow annually according to a normal random variable with mean 15% and standard deviation of 20%.

If we enter the market now our market share is equally likely to assume any value between 30% and 50%. Each year our market share will on average equal the previous year's market share with a standard deviation equal to 10% of the previous year's share. If we wait two years before entering our market share is equally likely to assume any value between 20% and 40%. Each year our market share will on average equal the previous year's market share with a standard deviation equal to 10% of the previous year's share.

Fell is trying to determine whether to enter the market now or wait two years and decide. Their goal is to maximize the mean NPV of their decision over 20 years. A 20% risk adjusted discount rate has been deemed appropriate for the project. What should Fell do?

**Solution**

To model this problem with RISKOPTIMIZER we will use two Adjustable cells:

- A 0-1 cell that indicates whether we enter now or not. A "1" will indicate Year 1 entry. A "0" will indicate no Year 1 entry.

- An integer cell that represents a threshold point T for Year 2 demand that drives our Year 3 entry decision. If we have not already entered and Year 2 demand is at least T we enter during Year 3, otherwise we do not enter during Year 3.

Our work is in file newtech.xls. See Figure 9.1.

**Figure 9.1**

	A	B	C	D	E	F	G	H	I	J
1	<b>New Technology</b>									
2										
3	Year	1	2	3	4	5	6	7	8	
4	Enter now?	0	0	0	0	0	0	0	0	
5	Year 2 cutoff for entry		6.62E+06							
6	Fixed cost	0		0	0	0	0	0	0	
7	Scenario	2	2	2	2	2	2	2	2	
8	Market Size	3739181.901	3793917	5121865	6893461	8068310	11444055	13871282	12439732	2396
9	Market Share	0	0	0	0	0	0	0	0	
10	Unit Margin	60	60	60	60	60	60	60	60	
11	Profit	0	0	0	0	0	0	0	0	
12	NPV	0.00E+00								
13										
14										
15										
16										

Through judicious use of IF statements we will ensure that the spreadsheet models the entry decision to be consistent with the previous definition of our adjustable cells.

**Step 1:** In B4 enter a trial value (0 or 1) for the Year 1 entry decision. In cell C5 enter a trial value for the Year demand cutoff which will generate a year 3 entry. In cell D4 we indicate if an entry occurs during Year 3 with the formula

$$=IF(AND(B4=0,C8>=C5),1,0).$$

This formula ensures that Year 3 entry occurs if and only if Year 1 entry does not occur and Year 2 demand exceeds cutoff.

**Step 2:** In cells B6 and D6 we generate the entry cost (if any). Year 1 entry cost is generated in cell B6 with the formula

$$=IF(B4=1,3000000000,0).$$

Year 3 entry cost is generated in cell D6 with the formula

$$=IF(D4=1,3500000000,0).$$

Entry cost is \$0 in all other years.

**Step 3:** In row 7 we generate the demand scenario. Of course, when we make our decisions we do not know which demand scenario has transpired. In cell B7 we generate the demand scenario (1 = High, 2 = Medium, 3 = Low) with the formula

*=RiskDiscrete({1,2,3},{0.1,0.5,0.4}).*

Entering

*=B7*

in C7 and copying that formula to D7:U7 ensures that the same demand scenario is used to generate market size for each year.

**Step 4:** In cell B8 we generate Year 1 market size with the formula

*=RiskNormal(5000000,1000000).*

In cell C8 we generate Year 2 market size with the formula

*=IF(B7=1,B8\*RiskNormal(2,0.2),IF(B7=2,B8\*RiskNormal(1.2,0.15),B8\*RiskNormal(0.95,0.15))).*

This formula grows the market according to the growth rate associated with the demand scenario that has actually occurred. Copying this formula to D8:G8 generates demand for years 3-6. In cell H8 we generate Year 7 demand with the formula

*=IF(G7=1,G8\*RiskNormal(1.15,0.2),IF(G7=2,G8\*RiskNormal(1.05,0.15),G8\*RiskNormal(0.8,0.15))).*

Copying this formula to I8:U8 models the slowdown in market growth due to product maturity.

**Step 5:** In cell B9 we use the =RISKUNIFORM function to model our Year 1 market share (assuming a year 1 entry).

*=IF(\$B4=1,RiskUniform(0.3,0.5),0).*

Our Year 2 market share is computed in C9 with the formula

*IF(\$B4=1,RiskNormal(1,0.1)\*B9,0).*

Our Year 3 market share is computed in D9 with the formula



$=IF(\$B4=1,C9*RiskNormal(1,0.1),IF(\$D4=1,RiskUniform(0.2,0.4),0)).$

This formula ensures that if Year 1 entry occurred, we modify our Year 2 market share. If Year 3 entry occurs, our share is equally likely to be between 30% and 50%. Otherwise we have no market share.

Finally, copying from E9 to F9:U9 the formula

$=IF(\$B4=1,D9*RiskNormal(1,0.1),IF(\$D4=1,D9*RiskNormal(1,0.1),0)).$

generates our market share for Years 4-20.

**Step 6:** In B10:U10 we compute the unit profit margin associated with the demand scenario by copying from B10 to C10:U10 the formula

$=IF(B7=1,90,IF(B7=2,60,40)).$

**Step 7:** In B11:U11 we compute each year's profit by copying the formula

$=B10*B9*B8-B6$

from B11 to C11:U11. For each year this formula computes


(market size)\*(market share)\*(unit profit contribution) - building costs.

**Step 8:** In cell B12 we compute our 20-year NPV (assuming end of year cash flows and a 20% discount rate) with the formula

$=NPV(0.2,B11:U11).$

**Step 9:** We are now ready to use RISKOPTIMIZER to determine an optimal decision strategy.

Our Settings box follows:



We choose to maximize mean NPV (cell B12) by adjusting our Year 1 entry decision (a 0-1 adjustable cell in B4) and our Year 2 entry threshold (cell C5). We constrain our Year 2 entry threshold to be an integer between 0 and 20,000,000. RISKOPTIMIZER recommends not entering now. If Year 2 demand is at least 6.62 million, then Fell should enter the market. Otherwise do not enter the market. An expected NPV of around \$1.53 billion is associated with this strategy.

## Section 11-Valuing a Cancellation Option

When companies purchase a quantity of a product, they often have an option to cancel (at a later date) part of the order if industry demand is less than expected. As a specific example, airlines often have an option to cancel part of an order for airplanes if an industry downturn occurs. The following example shows how to determine the value of a cancellation option.

### Example 1

Flyaway Airlines can order airplanes for \$5 million (payable when the planes are delivered in two years). Currently (year 0) they have 110 planes that will last for 11 years. Currently, the airline needs 110 planes. The number of planes needed is expected to grow according to a normal distribution by an average of 5% per year with a standard deviation of 10%. If a plane is needed during a year it generates \$2 million in profits

during the year. Assume a 15% discount rate is appropriate for this situation and we should consider cash flows occurring during years 0-10. Assume all cash flows occur at the end of the year.

- How many planes should we order today?
- Suppose that upon observing industry demand two years from now we can cancel up to half the planes ordered. How many planes should we order, and how should we plan to exercise the cancellation option? By how much has the cancellation option improved our NPV?

**Part (a) Solution**

Our Part (a) solution is in the sheet air3.xls and Figure 1.

**Figure 1**

	A	B	C	D	E	F	L
1	<b>Plane orders</b>						
2							
3	Profit if used	2.00E+06					
4	Plane cost	5.00E+06					
5	discount rate						
6	Order Size	23					
7	Year	0	1	2	3	4	10
8	Planes on hand	110	110	133	133	133	133
9	Planes needed	105	117	128	134	121	121
10	Number coming			23	23	23	23
11	Profit from planes			-7.90E+07	4.60E+07	2.20E+07	2.20E+07
12							
13	NPV	\$34,468,761.70					
14							
15							1.7
16							
17							
18							

**Step 1:** Enter our annual profit for a used plane and order cost per plane in B3 and B4, respectively.

**Step 2:** Enter a trial number of planes to order in cell B6.

**Step 3:** In B8 and C8 enter the number of planes currently available (110).

**Step 4:** In D8:L8 we compute the number of planes available for years 3-10. For years 3-10 the number of planes is simply 110 + (Planes Ordered). In D8 we compute this number with the formula

$$=C8+B6.$$

Copying the formula

= D8

from E8 to F8:L8 computes the number of planes available each year.

**Step 5:** In B9:L9 we compute the (random) number of planes needed each year. In B9 we enter 110. To generate the Year 2 number of planes needed we copy from C9 to D9:L9 the formula

= *ROUND(RiskNormal(1.05\*B9,0.1\*B9),0)*.

**Step 6:** In D10:L10 we enter the number of planes ordered by copying from D10 to E10:L10 the formula

=*\$B\$6*.

**Step 7:** In cell D11 we compute the Year 2 profit generated from the new planes. We first compute the cost of ordering the planes (just multiply number ordered times cost per plane. Next we observe that if demand for planes is 110 or less, no new planes will generate any profit. Otherwise the number of planes generating \$2 million in profit equals the minimum of number of planes ordered and amount needed planes exceeds 110. For example, if we order 23 planes and 130 are needed, 20 new planes will be used while if demand were 140 all 23 new planes would be used. Putting this together in cell D11 we enter the formula

=*-\$B\$4\*D10+IF(D9<B8,0,MIN(D10,D9-\$B\$8)\*\$B\$3)*.

**Step 8:** In cells E11:L11 we compute the profit earned by the new planes during years 3-10. Again remember that if demand for planes is 110 or less, no new planes will generate any profit. Otherwise the number of planes generating \$2 million in profit equals the minimum of number of planes ordered and amount needed planes exceeds 110. By copying from E11 to F11:L11 the formula


=*IF(E9<\$B\$8,0,MIN(E10,E9-\$B\$8)\*\$B\$3)*

we compute the profit earned by the new planes during years 3-10.

**Step 9:** In cell B13 we compute the NPV (in beginning of Year 2 \$s) of the profits generated by the new planes with the formula

= *NPV(0.15,D11:L11)*.

**Step 10:** We are now ready to use RISKOPTIMIZER to determine the number of planes that maximizes our expected NPV. Our settings box is as follows:



We simply choose number of planes ordered (B6) to maximize expected NPV (cell B13). We constrain the number of planes ordered to be an integer between 1 and 100. RISKOPTIMIZER recommends ordering 23 planes to maximize expected NPV. A mean NPV of \$36.7 million (Year 2 dollars) is obtained.

### Part (b) Solution

We now allow Flyaway Airlines to cancel up to half their order if Year 2 demand is less than anticipated. Our work is in file air2.xls and Figure 2.

**Figure 2**

	A	B	C	D	E	F	L
1	<b>Plane orders</b>						
2							
3	Profit if used	2.00E+06					
4	Plane cost	5.00E+06					
5	discount rate						
6	Order Size	35					
7	Year	0	1	2	3	4	10
8	Planes on hand	110	110	145	145	145	145
9	Planes needed	105	115	132	130	124	122
10	Cancel key			0.1733866			
11	Cancelled			5			
12	Number coming			30	30	30	30
13	Profit from planes			-1.06E+08	4.00E+07	2.80E+07	2.40E+07
14							
15	NPV	(\$20,194,021.27)		Cancelled	%age	slope	
16				60	0.5	0	
17				70	0.5	0	1.7
18				80	0.5	0	
19				90	0.5	0	
20				100	0.5	0	
21				110	0.5	-0.018675	
22				120	0.31324981	-0.013986	
23				130	0.17338661	-0.017339	

We will assume Flyaway's cancellation strategy will be keyed by the percentage of the order cancelled when Year 2 demand is 60, 70, 80, 90, 100, 110, 120, or 130 planes. For other values of demand, the cancellation percentage will be obtained by linear interpolation, subject to the constraint that the percentage of the order cancelled must be non-negative. The cancellation percentages keying the strategy are in cells E16:E23. In F16:F23 we determine the "slope" of the cancellation percentage. To illustrate how the strategy is implemented, if Year 2 demand is for 115 planes then  $.5 + (115-110)*(-0.018675) = 40.7\%$  of the ordered will be cancelled. The "interpolation strategy" may not be optimal but it should be close to the optimal strategy. The key to implementing the interpolation strategy is a clever use of lookup tables. We proceed as follows:

**Step 1:** Insert two rows below row 9 in the spreadsheet.

**Step 2:** In D16:D23 enter the Year 2 demand values (60-130) which key the interpolation. Then in E16:E23 enter trial values of cancellation percentages for each demand value. Next in F16:F23 compute the "slope" of the cancellation percentage between key points by copying from F16 to F17:E23 the formula

$$=(E17-E16)/10.$$

Our cancellation percentage will now be a piecewise linear function of the Year 2 demand.

**Step 3:** In cell D10 we determine the cancellation percentage keying our interpolation with the formula

$=VLOOKUP(D9, \$D\$16: \$F\$23, 2).$

**Step 4:** In cell D11 we compute the number of planes cancelled for the actual number of planes needed in Year 2 with the formula

$=MAX(0, ROUND(B6*(D10 + (D9 - VLOOKUP(D9, \$D\$16: \$F\$23, 1)) * VLOOKUP(D9, \$D\$16: \$F\$23, 3)), 0)).$

The = MAX insures that the number of cancelled planes is always non-negative. The rest of the formula multiplies the number of planes ordered times the cancellation percentage obtained by interpolation and rounds that number to the nearest integer.

**Step 5:** In D12:L12 we compute the number of planes received (after cancellation). First enter in D12 the formula

$=\$B\$6 - D11.$

Copying the formula

$=D12$

from E12:L12 recopies this number of planes received to Years 3-10.

**Step 6:** In D13 we compute our Year 2 profit from planes received After noting that we only pay for planes received and can only use planes received we determine Year 2 profit in E12 with the formula

$=-\$B\$4 * D12 + IF(D9 < B8, 0, MIN(D12, D9 - \$B\$8) * \$B\$3).$

**Step 7:** In cells E13:L13 we compute the profit from planes received during years 2-10 by copying the formula


$=IF(E9 < \$B\$8, 0, MIN(E12, E9 - \$B\$8) * \$B\$3)$

from E13 to F13:L13.

**Step 8:** In cell B15 compute (in beginning of Year 2 dollars) the NPV of all profits earned from the planes received with the formula

$=NPV(0.15, D13: L13).$

**Step 9:** We are now ready to use RISKOPTIMIZER to determine the optimal number of planes to order and the optimal cancellation policy. Here is our RISKOPTIMIZER settings:



Our goal is to maximize mean NPV (cell B15). Our adjustable cells are the number of planes ordered ( cell B6) and the cancellation percentages for an observed year 2 demand for planes of 60-130 planes (cells E16:E23). We constrain the number of planes ordered to be an integer between 0 and 80. Each cancellation percentage is constrained to be between 0 and .5.

RISKOPTIMIZER recommends ordering 35 planes. If Year 2 demand is 110 or less, cancel half the order. If demand is 120, cancel 31% of the order. If demand is 130, cancel 17% of the order. Our mean NPV is \$43.4 million. Thus the option to cancel has enabled us to order more planes (due to our increased flexibility) and has increased our average profit by around \$6.7 million.



## References

- Brealey, R., and Myers, S., *Principles of Corporate Finance*, Prentice-Hall, 1996.
- Hull, J. *Options Futures and Derivative Securities*, Prentice-Hall, 1997.
- Luenberger, D., *Investment Science*, Oxford Press, 1997.
- Nichols, N., "Scientific Management at Merck: An Interview with CFO Judy Lewent, *Harvard Business Review*, Vol. 72, No. 1, pages 89-94, 1994.
- Trigeorgis, L., *Real Options*, MIT Press, 1996.
- Winston, W.L., *Decision-Making Under Uncertainty Using RISKOPTIMIZER*, Palisades, 1999.
- Winston, W.L., *Financial Models Using Simulation and Optimization*, Palisades, 1998.